

Non-Isomorphic Real Simple Lie Algebras of the Same Complex Type and Character

Synopsis

- Complex simple Lie algebras are classified by their root types -- these are labeled as $A_r, B_r, C_r, D_r, E_6, E_7, E_8, F_4, G_2$. The type of a real simple Lie algebra \mathfrak{g} is the root type of the associated complex algebra. The character δ of a real simple Lie algebra is the signature of its Killing form.
- For many root types, the character is sufficient to uniquely classify the corresponding real Lie algebras. Indeed in [1] (page 26) Cartan writes, "*Les groupes réels d'ordre r qui correspondent à une même type complexe d'ordre r se classent en général complètement d'après leur caractère.*" However, as Helgason [2] (page 537) writes, one should not take this statement to be literally true – there are a few cases where the character does not suffice to distinguish all possible real forms.
- In Sections 1 and 2 we will show that the 2 real non-isomorphic [Lie algebras](#) $so^*(18)$ and $so(12, 6)$ (with isomorphic complexifications $so(18)$) have the same character $\delta = -9$.
- A more complete understanding of the isomorphism problem for real forms of a given complex type is provided by the following theorem (Helgason [2], page 517). We recall that if $\mathfrak{g} = \mathfrak{t} \oplus \mathfrak{p}$ is the [Cartan decomposition](#) of a real simple Lie algebra, then the Killing form is negative-definite on \mathfrak{t} and positive-definite on \mathfrak{p} . It follows that $\delta = \dim \mathfrak{p} - \dim \mathfrak{t}$.

Theorem 1. Let \mathfrak{g}_1 and \mathfrak{g}_2 be two real simple Lie algebras of the same complex type and with Cartan decompositions $\mathfrak{g}_1 = \mathfrak{t}_1 \oplus \mathfrak{p}_1$ and $\mathfrak{g}_2 = \mathfrak{t}_2 \oplus \mathfrak{p}_2$. Then \mathfrak{g}_1 and \mathfrak{g}_2 are isomorphic as real Lie algebras if and only if the reductive Lie algebras \mathfrak{t}_1 and \mathfrak{t}_2 are isomorphic.

- In Section 3 we show that the compact part of the Cartan decomposition for $so^*(18)$ is $\mathfrak{t} = \mathfrak{u}(1) \oplus \mathfrak{su}(9)$.
- In Section 4 we show that the compact part of the Cartan decomposition for $so(12, 6)$ is

$$\mathfrak{t} = su(4) \oplus so(12) = so(6) \oplus so(12).$$

1. The Real Lie Algebras of Type $A_r, B_r, C_r, D_r, 4 \leq r \leq 9$

Here are the characters for all the classical Lie algebras of rank $4 \leq r \leq 9$. We see that $so^*(18)$ and $so(12, 6)$ are the only classical Lie algebras of rank less than or equal to 9 which have the same type and character.

Type A:

A4 δ	sl (5) 4	su (4, 1) -24	su (3, 1) -8	su (2, 2) 0				
A5 δ	sl (6) 5	su (6) -35	su (5, 1) -15	su (4, 2) -3	su (3, 3) 1	su^* (6) -7		
A6 δ	sl (7) 6	su (7) -48	su (6, 1) -24	su (5, 2) -8	su (4, 3) 0			
A7 δ	sl (8) 7	su (8) -63	su (7, 1) -35	su (6, 2) -15	su (5, 3) -3	su (4, 4) 1	su^* (8) -9	
A8 δ	sl (9) 8	su (9) -80	su (8, 1) -48	su (7, 2) -24	su (6, 3) -8	su^* (5, 4) 0		

	6									
B 9	so (19)	so (18, 1)	so (17, 2)	so (16, 3)	so (15, 4)	so (14, 5)	so (13, 6)	so (12, 7)	so (11, 8)	so (10, 9)
δ	-17 1	-135	-103	-75	-51	-31	-15	-3	5	9

Type C:

C4	sp (8)	sp(6, 2)	sp(4, 4)	sp (8, R)		
δ	-36	-12	-4	4		
C5	sp (10)	sp(8, 2)	sp(6, 4)	sp (10, R)		
δ	-55	-23	-7	5		
C6	sp (12)	sp (10, 2)	sp(8, 4)	sp (6, 6)	sp (12, R)	
δ	-78	--38	-14	-6	6	
C7	sp (14)	sp (12, 2)	sp (10, 4)	sp (8, 6)	sp (14, R)	
δ	-105	-57	-25	-9	7	
C8	sp (16)	sp (14, 2)	sp (12, 4)	sp (10, 6)	sp(8, 8)	sp (16, R)
δ	-136	-80	-40	-16	-8	8
C9	sp (18)	sp (16,	sp (14,	sp (12,	sp (10,	sp (18,

δ	-171	2) - 107	4) -59	6) -27	8) -11	R) 9
----------	------	----------------	-----------	-----------	-----------	---------

Type D:

D4	so δ (8) -28	so (7,1) -14	so (6, 2) -4	so (5, 3) 2	so (4, 4) 4	so*	(8) -4					
D5	so δ (10) -45	so (9,1) -27	so (8, 2) -13	so (7, 3) -3	so (6, 4) 3	so	so* (10) -5					
D6	so δ (12) -66	so (11, 1) -44	so (10, 2) -26	so (9, 3) -12	so (8, 4) -2	so	so (7, 5) 4	so (6, 6) 6	so -6	so* (12)		
D7	so δ (14) -91	so (13, 1) -65	so (12, 2) -43	so (11, 3) -25	so (10, 4) -11	so	so (9, 5) -1	so (8, 6) 5	so (7, 7) 7	so* (14) -7		
D8	so δ (16) -12 0	so (15, 1) -90	so (14, 2) -64	so (13, 3) -42	so (12, 4) -24	so	so (11, 5) -10	so (10, 6) 0	so (9, 7) 6	so(8, 8) 8	so -8	so* (16)
D9	so δ (18)	so (17,	so (16,	so (15,	so (14,	so (13,	so (12,	so (11,	so (10,	so (9,9)	so*	(18)

)	1)	2)	3)	4)	5)	6)	7)	8)	9	-9
-15	-119	-89	-63	-41	-23	-9	1	7		
3										

Details

Here are some simple programs used to generate the characters for the above tables.

```
[alg > with(DifferentialGeometry) : with(LieAlgebras) :
```



For example, the command `Adelta(r)` calculates the characters for the real forms of A_r , namely, $sl(r+1)$, $su(r+1)$, $su(r,1)$, ...

:

```
[alg > Adelta(5) ;
```

$5, [5, -35, -15, -3, 1, -7]$ (2.1.1)

```
[alg > Bdelta(5) ;
```

$5, [55, -35, -19, -7, 1, 5]$ (2.1.2)

2. Simple roots and Satake Diagrams for $so^*(18)$ and $so(12, 6)$

In this section we initialize the Lie algebras $so^*(18)$ and $so(12, 6)$ and calculate their properties. We check that the characters of each are equal to -9. From the simple roots we are easily able to confirm that these Lie algebras are not isomorphic.

```
[> with(DifferentialGeometry) : with(LieAlgebras) :
```

Part 1 - so*(18):

Retrieve the [structure equations](#) for so*(18) and initialize the algebra.

```
[> LD1 := SimpleLieAlgebraData("so*(18)", alg1):  
[> DGsetup(LD1, [e], [o]):
```

Calculate the [properties](#) of so*(18) (this will take several minutes to complete).

```
[alg1 > Properties1 := SimpleLieAlgebraProperties(alg1):
```

Here is the [Cartan decomposition](#).

```
[alg1 > T1, P1 := Properties1:-CartanDecomposition;  
T1, P1 := [e1, e2, e3, e4, e5, e6, e7, e8, e9, e10, e11, e12, e13, e14, e15, e16, e17, e18, e19, e20, e21, e22, e23, e24, e25, e26, e27,  
e28, e29, e30, e31, e32, e33, e34, e35, e36, e73, e74, e75, e76, e77, e78, e79, e80, e81, e82, e83, e84, e85, e86, e87, e88, e89,  
e90, e91, e92, e93, e94, e95, e96, e97, e98, e99, e100, e101, e102, e103, e104, e105, e106, e107, e108, e109, e110, e111, e112,  
e113, e114, e115, e116, e117], [e37, e38, e39, e40, e41, e42, e43, e44, e45, e46, e47, e48, e49, e50, e51, e52, e53, e54, e55,  
e56, e57, e58, e59, e60, e61, e62, e63, e64, e65, e66, e67, e68, e69, e70, e71, e72, e118, e119, e120, e121, e122, e123, e124,  
e125, e126, e127, e128, e129, e130, e131, e132, e133, e134, e135, e136, e137, e138, e139, e140, e141, e142, e143, e144,  
e145, e146, e147, e148, e149, e150, e151, e152, e153]
```

(3.1)

We see that the character of so*(18) is -9.

```
[alg1 > nops(T1), nops(P1);  
81, 72
```

(3.2)

Here are the [simple roots](#) for so*(18).

```
[alg1 > SimRts1 := Properties1:-SimpleRoots;
```

$$\text{SimRts1} := \left[\begin{array}{c} \left[\begin{array}{c} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{array} \right], \left[\begin{array}{c} 1 \\ -1 \\ 0 \\ 0 \\ -1 \\ 0 \\ 0 \\ 0 \end{array} \right], \left[\begin{array}{c} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 2I \\ 0 \\ 0 \end{array} \right], \left[\begin{array}{c} 0 \\ 1 \\ -1 \\ 0 \\ -I \\ 0 \\ 0 \\ 0 \end{array} \right], \left[\begin{array}{c} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 2I \\ 0 \end{array} \right], \left[\begin{array}{c} 0 \\ 0 \\ 1 \\ -1 \\ -I \\ -I \\ 0 \\ 0 \end{array} \right], \left[\begin{array}{c} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 2I \end{array} \right], \left[\begin{array}{c} 0 \\ 0 \\ 0 \\ 1 \\ 0 \\ 0 \\ -I \\ -I \end{array} \right], \left[\begin{array}{c} 0 \\ 0 \\ 0 \\ 1 \\ 0 \\ 0 \\ -I \\ -I \end{array} \right], \left[\begin{array}{c} 0 \\ 0 \\ 0 \\ 1 \\ 0 \\ 0 \\ -I \\ I \end{array} \right] \end{array} \right] \quad (3.3)$$



We see that there are 4 pure imaginary roots so there are 4 black dots in the [SatakeDiagram](#). The [Satake associate](#) of root 8 is root 9.

```
alg1 > SatakeAssociate(SimRts1[8], SimRts1);
```

$$\left[\begin{array}{c} 0 \\ 0 \\ 0 \\ 1 \\ 0 \\ 0 \\ 0 \\ -I \\ I \end{array} \right] \quad (3.4)$$

These 2 facts are coded in the Satake diagram for $so^*(18)$.

```
alg1 > SatakeDiagram("so*(18)");
```

Now we turn to the second Lie algebra.

Part 2 - so(12, 6):

Retrieve the [structure equations](#) for so(12, 6) and initialize the algebra.

```
[> LD2 := SimpleLieAlgebraData("so(12, 6)", alg2):
[> DGsetup(LD2, [f], [o]):
```

Calculate the [properties](#) of so(12, 6) (this will take several minutes to complete).

```
[alg2 > Properties2 := SimpleLieAlgebraProperties(alg2):
```

Here is the [Cartan decomposition](#).

```
alg2 > T2, P2 := Properties2:-CartanDecomposition;
```

```
T2, P2 := [f2 - f7, f3 - f13, f4 - f19, f5 - f25, f6 - f31, f9 - f14, f10 - f20, f11 - f26, f12 - f32, f16 - f21, f17 - f27, f18
- f33, f23 - f28, f24 - f34, f30 - f35, f37 + f52, f38 + f53, f39 + f54, f40 + f55, f41 + f56, f42 + f57, f43 + f58, f44
+ f59, f45 + f60, f46 + f61, f47 + f62, f48 + f63, f49 + f64, f50 + f65, f51 + f66, f67 + f103, f68 + f104, f69 + f105, f70
+ f106, f71 + f107, f72 + f108, f73 + f109, f74 + f110, f75 + f111, f76 + f112, f77 + f113, f78 + f114, f79 + f115, f80
+ f116, f81 + f117, f82 + f118, f83 + f119, f84 + f120, f85 + f121, f86 + f122, f87 + f123, f88 + f124, f89 + f125, f90
+ f126, f91 + f127, f92 + f128, f93 + f129, f94 + f130, f95 + f131, f96 + f132, f97 + f133, f98 + f134, f99 + f135, f100
+ f136, f101 + f137, f102 + f138, f139, f140, f141, f142, f143, f144, f145, f146, f147, f148, f149, f150, f151, f152, f153], [f1,
f2 + f7, f3 + f13, f4 + f19, f5 + f25, f6 + f31, f8, f9 + f14, f10 + f20, f11 + f26, f12 + f32, f15, f16 + f21, f17 + f27, f18
+ f33, f22, f23 + f28, f24 + f34, f29, f30 + f35, f36, f37 - f52, f38 - f53, f39 - f54, f40 - f55, f41 - f56, f42 - f57, f43
- f58, f44 - f59, f45 - f60, f46 - f61, f47 - f62, f48 - f63, f49 - f64, f50 - f65, f51 - f66, f67 - f103, f68 - f104, f69
- f105, f70 - f106, f71 - f107, f72 - f108, f73 - f109, f74 - f110, f75 - f111, f76 - f112, f77 - f113, f78 - f114, f79
- f115, f80 - f116, f81 - f117, f82 - f118, f83 - f119, f84 - f120, f85 - f121, f86 - f122, f87 - f123, f88 - f124, f89
- f125, f90 - f126, f91 - f127, f92 - f128, f93 - f129, f94 - f130, f95 - f131, f96 - f132, f97 - f133, f98 - f134, f99
- f135, f100 - f136, f101 - f137, f102 - f138]
```

(3.5)

We see that the character of $\mathfrak{so}(12, 6)$ is -9, which is the same as $\mathfrak{so}^*(18)$.

```
alg2 > nops(T2), nops(P2);
```

81, 72

(3.6)

Here are the [simple roots](#) for $\mathfrak{so}(12, 6)$.



```
alg2 > SimRts2 := Properties2:-SimpleRoots;
```

$$\text{SimRts2} := \left[\begin{array}{c} \left[\begin{array}{c} 1 \\ -1 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{array} \right], \left[\begin{array}{c} 0 \\ 1 \\ -1 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{array} \right], \left[\begin{array}{c} 0 \\ 0 \\ 1 \\ -1 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{array} \right], \left[\begin{array}{c} 0 \\ 0 \\ 0 \\ 1 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{array} \right], \left[\begin{array}{c} 0 \\ 0 \\ 0 \\ 0 \\ -1 \\ 0 \\ 0 \\ 0 \\ 0 \end{array} \right], \left[\begin{array}{c} 0 \\ 0 \\ 0 \\ 0 \\ 1 \\ -1 \\ 0 \\ -I \\ 0 \end{array} \right], \left[\begin{array}{c} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ I \\ -I \\ I \\ 0 \end{array} \right], \left[\begin{array}{c} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ I \\ I \\ -I \end{array} \right], \left[\begin{array}{c} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ I \end{array} \right] \end{array} \right] \quad (3.7)$$

We see that there are 3 pure imaginary roots (the last 3 simple roots) so there are 3 black dots in the Satake diagram. *This suffices to show that the Lie algebras $so^*(18)$ and $so(12, 6)$ are not isomorphic.*

The Satake diagram for $so(12, 6)$ is:

```
alg1 > SatakeDiagram("so(12, 6)");
```



3. Classification of the Compact Part of the Cartan Decomposition for $so^*(18)$

Initialize the compact part **T1** of the Cartan decomposition of $so^*(18)$ as a Lie algebra in its own right.

(4.1)

```
[alg2 > LDC1 := LieAlgebraData(T1, Compact1):
[alg2 > DGsetup(LDC1, [x], [o]);
```

Lie algebra: Compact1

(4.2)

Since the compact part of the Cartan decomposition is always reductive but perhaps not semi-simple, we calculate the [Levi decomposition](#). This takes several minutes to calculate so we store the result.

```
[Compact1 > #Levi1 := LeviDecomposition(Compact1);
```

```
[Compact1 > Levi1 := [[x37+x46+x54+x61+x67+x72+x76+x79+x81], [x1, x2, x3, x4, x5,
x6, x7, x8, x9, x10, x11, x12, x13, x14, x15, x16, x17, x18, x19, x20,
x21, x22, x23, x24, x25, x26, x27, x28, x29, x30, x31, x32, x33, x34,
x35, x36, x37-x81, x38, x39, x40, x41, x42, x43, x44, x45, x46-x81, x47,
```

```

x48, x49, x50, x51, x52, x53, x54-x81, x55, x56, x57, x58, x59, x60, x61
-x81, x62, x63, x64, x65, x66, x67-x81, x68, x69, x70, x71, x72-x81,
x73, x74, x75, x76-x81, x77, x78, x79-x81, x80]];

```

```

Levi1 := [[x37 + x46 + x54 + x61 + x67 + x72 + x76 + x79 + x81], [x1, x2, x3, x4, x5, x6, x7, x8, x9, x10, x11, x12, x13, x14,
x15, x16, x17, x18, x19, x20, x21, x22, x23, x24, x25, x26, x27, x28, x29, x30, x31, x32, x33, x34, x35, x36, x37 - x81, x38, x39,
x40, x41, x42, x43, x44, x45, x46 - x81, x47, x48, x49, x50, x51, x52, x53, x54 - x81, x55, x56, x57, x58, x59, x60, x61 - x81,
x62, x63, x64, x65, x66, x67 - x81, x68, x69, x70, x71, x72 - x81, x73, x74, x75, x76 - x81, x77, x78, x79 - x81, x80]]

```

(4.3)

We see there is 1-dimensional center.

```

Compact1 > Levi1[1];
[x37 + x46 + x54 + x61 + x67 + x72 + x76 + x79 + x81]

```

(4.4)

Initialize the semi-simple part of the Levi decomposition.

```

Compact1 > LDCss1 := LieAlgebraData(Levi1[2], Compact1ss):
Compact1 > DGsetup(LDCss1, [y], [o]);
Lie algebra: Compact1ss

```

(4.5)

To classify the Lie algebra **Compact1ss**, we need a Cartan subalgebra. Because this algebra is quite large, it helps to first identify an abelian Lie algebra **A** of semi-simple (that is, diagonalizable) elements in the algebra and then look for a Cartan subalgebra containing **A**.

```

Compact1ss > Query(Adjoint(y1), "Diagonalizable");
true

```

(4.6)

```

Compact1ss > GNS1 := GeneralizedNullSpace([y1]);
GNS1 := [y1, y16, y17, y18, y19, y20, y21, y22, y23, y24, y25, y26, y27, y28, y29, y30, y31, y32, y33, y34, y35, y36, y37 + y46,
y54, y55, y56, y57, y58, y59, y60, y61, y62, y63, y64, y65, y66, y67, y68, y69, y70, y71, y72, y73, y74, y75, y76, y77, y78, y79,
y80]

```

(4.7)

```

Compact1ss > A := [y1, y16];
A := [y1, y16]

```

(4.8)

```

Compact1ss > Query(A, "Abelian");
true

```

(4.9)

```

Compact1ss > CSA1 := CartanSubalgebra(contains = A);

```

(4.10)

$$CSA1 := [y1, y16, y27, y37 + y46, y54 + y61, y67 + y72, y76, y79] \quad (4.10)$$

Now classify the Lie algebra.

```
Compact1ss > infolevel[ClassifyComplexSemiSimpleLieAlgebra] := 2;
infolevel DifferentialGeometry:-LieAlgebras:-ClassifyComplexSemiSimpleLieAlgebra := 2
```

(4.11)

```
Compact1ss > ClassifyComplexSemiSimpleLieAlgebra(CSA1);
Step 1. Find the root space decomposition
Step 2. Find the positive and simple roots
Step 3. Find the Cartan matrix
Step 4. Put the Cartan matrix in standard form
"A8"
```

(4.12)

We conclude that the compact part of the Cartan decomposition for $so^*(18)$ is $\mathfrak{t}_1 = \mathfrak{u}(1) \oplus \mathfrak{su}(9)$.

4. Classification of the Compact Part of the Cartan Decomposition for $so(12, 6)$

Initialize the compact part T2 of the Cartan decomposition of $so(12, 6)$.

```
Compact1ss > LDC2 := LieAlgebraData(T2, Compact2);
alg2 > DGsetup(LDC2, [z], [o]);
Lie algebra: Compact2
```

(5.1)

This time the compact part is semi-simple and so there is no need to calculate the Levi decomposition. We need a Cartan subalgebra and, as before, it helps to first identify an abelian Lie algebra \mathfrak{A} of semi-simple (that is, diagonalizable) elements in the algebra.

```
Compact2 > Query(Adjoint(z1), "Diagonalizable");
true
```

(5.2)

```
Compact2 > GNS2 := GeneralizedNullSpace([z1]);
GNS2 := [z1, z10, z11, z12, z13, z14, z15, z16, z25, z26, z27, z28, z29, z30, z43, z44, z45, z46, z47, z48, z49, z50, z51, z52, z53,
z54, z55, z56, z57, z58, z59, z60, z61, z62, z63, z64, z65, z66, z67, z68, z69, z70, z71, z72, z73, z74, z75, z76, z77, z78, z79,
z80, z81]
```

(5.3)

```
Compact2 > A := [z1, z10];  
A := [z1, z10] (5.4)
```

```
Compact2 > Query(A, "Abelian");  
true (5.5)
```

```
Compact2 > CSA2 := CartanSubalgebra(contains = A);  
CSA2 := [z1, z10, z15, z16, z25, z30, z67, z76, z81] (5.6)
```

```
Compact2 > ClassifyComplexSemiSimpleLieAlgebra(CSA2);  
Step 1. Find the root space decomposition  
Step 2. Find the positive and simple roots  
Step 3. Find the Cartan matrix  
Step 4. Put the Cartan matrix in standard form  
"6A3" (5.7)
```

We conclude that the compact part of the Cartan decomposition for $so(12, 6)$ is $t_2 = su(4) \oplus so(12) = so(6) \oplus so(12)$.

Highlighted Commands

[ClassifyComplexSemiSimpleLieAlgebra](#), [GeneralizedNullSpace](#), [SimpleLieAlgebraProperties](#), [SatakeDiagram](#)

Release Notes

• This worksheet was compiled with Maple 17 and DG release USU1, available by request from ian.anderson@usu.edu.

References

1. E. Cartan, *Les groupes réels simples finis et continus*, Ann. Sci. École Norm. Sup. 31 (1914), 263-355.
2. S. Helgason, *Differential Geometry, Lie Groups and Symmetric Spaces*, Pure and Applied Mathematics **80**, Academic Press (1978).

Author

Ian M. Anderson
Department of Mathematics and Statistics
Utah State University
January 8, 2015
ian.anderson@usu.edu

